

BÀI TẬP VÀ THỰC HÀNH 7

A. Mục đích, yêu cầu

- Tiếp tục củng cố cho học sinh những khái niệm về chương trình con: thủ tục, hàm, tham số giá trị, tham số biến, tham số thực sự, biến toàn cục, biến cục bộ.
- Rèn luyện cho học sinh kỹ năng lập trình giải quyết một số bài toán đơn giản.
- Tổ chức những hoạt động trong phòng máy để học sinh có được các kỹ năng cơ bản trong việc tổ chức và sử dụng chương trình con trong lập trình.

B. Những điểm cần lưu ý và gợi ý tổ chức thực hành

Về câu a

- Học sinh chuẩn bị ở nhà, đầu giờ thực hành giáo viên ra một số câu hỏi kiểm tra việc chuẩn bị của học sinh và chỉnh sửa, giải đáp.
- Mục đích của câu này là:
 - Làm cho học sinh hiểu rõ cách sử dụng kiểu bản ghi để mô tả một tam giác. Tuy có thể sử dụng từng cặp biến đơn để thể hiện hai toạ độ của mỗi đỉnh của tam giác, nhưng việc sử dụng kiểu bản ghi *Diem* cho người lập trình nhìn một điểm gọn trong một biến với hai toạ độ thành phần luôn ràng buộc với nhau trong biến này. Cũng với cách nhìn như vậy khi coi tam giác là một bản ghi gồm ba trường, mỗi trường cho thông tin về một đỉnh của tam giác, người lập trình có được cách nhìn gọn gàng, nhất quán và có cấu trúc về thông tin các đối tượng mà chương trình cần xử lý.

Điều này tạo điều kiện cho việc kiến thiết những chương trình sáng sủa, dễ hiểu, ít nhầm lẫn.

- Làm học sinh hiểu cụ thể yêu cầu của từng bài toán con tương ứng với mỗi chương trình con sẽ được tìm hiểu ở câu *b*. Học sinh cần biết dữ liệu input, output của mỗi bài toán con này, từ đó hiểu được rằng nếu tổ chức một chương trình con thì nên dùng loại nào (thủ tục hay hàm), nên dùng những tham số thuộc loại gì (tham số giá trị hay tham số biến), giá trị của hàm, giá trị của các tham số thuộc kiểu dữ liệu nào? Một lần nữa học sinh cần thấy rằng việc sử dụng biến kiểu bản ghi thể hiện điểm, thể hiện tam giác sẽ làm việc khai báo các tham số hình thức của chương trình con ngắn gọn, sáng sủa, dễ hiểu hơn.

Về câu b

- Học sinh đọc và tìm hiểu trước chương trình ở nhà.
- Mục đích câu này là cung cấp cho học sinh và yêu cầu học sinh hiểu được các chương trình con giải quyết một số bài toán đơn giản liên quan đến tam giác khi biết tọa độ các đỉnh: Tính độ dài cạnh, chu vi, diện tích, kiểm tra các tính chất (đều, cân, vuông).
- Chú ý rằng hàm tính khoảng cách giữa hai điểm P, Q thuộc kiểu real, cho giá trị gần đúng. Do vậy, kết quả tính độ dài cạnh tam giác, chu vi và diện tích tam giác cũng là gần đúng. Việc kiểm tra để khẳng định tính chất (đều, cân, vuông) của một tam giác với thủ tục *Tinhchat* cũng dựa trên những giá trị gần đúng và sai số chấp nhận được là $eps = 1.0E-6$.

Về câu c

- Tạo tình huống để học sinh vận dụng hiểu biết về các chương trình con ở câu *b* để giải quyết một bài toán khác liên quan đến việc xác định tính chất của tam giác khi biết tọa độ các đỉnh. Qua đó, phát triển kỹ năng lập trình nói chung và kỹ năng sử dụng chương trình con nói riêng cho học sinh. Một số điểm cần lưu ý:
 - Học sinh chưa được thực hành về kiểu dữ liệu tệp, do vậy giáo viên cần dành thời gian chỉ dẫn hoặc hướng dẫn cụ thể đoạn chương trình nhập dữ liệu từ tệp TAMGIAC.DAT.
 - Đối với dữ liệu nhập vào cho mỗi tam giác T , có thể dùng thủ tục *Tinhchat*($R, Deu, Can, Vuong$) để xác định tính chất của tam giác đó (đều, cân, vuông hay thường). Thủ tục *Tinhchat* có sử dụng thủ tục

$Daicanh(R, a, b, c)$ để tính chiều dài ba cạnh của tam giác khi biết tọa độ ba đỉnh. Do vậy, chương trình cần viết chỉ cần sử dụng hai thủ tục *Tinhchat* và *Daicanh* trong số các chương trình con đã đưa ra ở câu *b*. Hướng dẫn học sinh lưu chương trình ở câu *b* vào tệp với một tên khác (dùng **Save as**) để sửa đổi lại thành chương trình đáp ứng yêu cầu ở câu *c* này. Việc sửa đổi như vậy gồm xoá bớt các chương trình con không cần dùng, thêm một số khai báo cần thiết (các biến đếm số tam giác đều, số tam giác cân, số tam giác vuông), viết lại thân chương trình chính.

- Như đã biết ở câu *b*, tính chất tam giác được xác định dựa vào so sánh chiều dài các cạnh. Chiều dài các cạnh được tính gần đúng. Do vậy, việc xác định tính chất của một tam giác cũng gần đúng với sai số là 0,000001.
- Giáo viên nên chuẩn bị một số tệp input và output tương ứng, chép sẵn tệp dữ liệu input vào máy thực hành để học sinh có thể kiểm thử chương trình của mình.

Chương trình có thể được viết như sau:

```

uses crt;
type
    Diem = record
        x, y: real;
    end;
    Tamgiac = record
        A, B, C: Diem;
    end;
const eps = 1.0E-6;
var T: Tamgiac;
    D, Cn, V : boolean;
    N, i, deu, can, vuong: word;
    f : text;
function Kh_cach(P, Q: Diem): real;
begin
    Kh_cach:= sqrt((P.x-Q.x)*(P.x-Q.x) + (P.y-Q.y)*(P.y-Q.y));
end;
procedure Daicanh(var R: Tamgiac; var a, b, c: real);
begin
    a:= Kh_cach(R.B, R.C);
    b:= Kh_cach(R.A, R.C);
    c:= Kh_cach(R.A, R.B);
end;

```

```

procedure TinhChat (var R:Tamgiac;var Deu,Can,Vuong:boolean);
  var a, b, c: real;
  begin
    Deu:= false;
    Can:= false;
    Vuong:= false;
    DaiCanh(R, a, b, c);
    if (abs(a-b)<eps) and (abs(a-c)<eps) then
      begin
        Deu:= true;
        Can:= true;
      end
    else if (abs(a-b)<eps) or (abs(a-c)<eps) or
      (abs(b-c)<eps) then Can:= true;
    if (abs(a*a+b*b-c*c)<eps) or (abs(a*a+c*c-b*b)<eps)
      or (abs(b*b+c*c-a*a)<eps) then Vuong:= true;
  end;
begin
  deu:=0;
  can:=0;
  vuong:=0;
  assign(f, 'TAMGIAC.DAT');
  reset(f);
  readln(f,N);
  for i:=1 to N do
    begin
      readln(f, T.A.x, T.A.y,T.B.x,T.B.y,T.C.x,T.C.y);
      TinhChat(T,D,Cn,V);
      if D then deu:=deu+1
        else
          begin
            if Cn then can:=can+1;
            if V then vuong:= vuong+1;
          end;
        end;
    end;
  writeln('Co: ',deu,' tam giac deu');
  writeln('Co: ',can,' tam giac can (khong deu)');
  writeln('Co: ',vuong,' tam giac vuong');
  readln
end.

```